

Módulo 4

Funções e Subrotinas

Conteúdo

- Conceito e aplicação
 - O que são e para que servem as funções e subrotinas.

- Funções
 - Sintaxe e exemplos de aplicação.

- Subrotinas
 - Sintaxe e exemplos de aplicação.
 - Passagem de parâmetros por valor e referência.

Rotinas

- O que são rotinas?
 - Uma grande parte das linguagens de programação permitem agrupar instruções em blocos de instruções autónomos que podem ser chamados durante a execução do programa.
 - O mecanismo de “subcontratação” permite simplificar os programas estruturando o código em segmentos auto-suficientes.
 - Há dois tipos de rotinas:
 - Função – baseada no conceito matemático de função, em que o resultado é necessariamente um único valor.
 - Subrotina – implementando um mecanismo de “subcontratação” de tarefas, não sendo representável por um valor.

Funções - Sintaxe

- Sintaxe
 - Geral

```
Function <nome>([<argumento1>, ...]) as <tipo de dados>
    <bloco de instruções>
End Function
```
 - Exemplo

```
Function Inverso (Byval num as Single) as Single
    Inverso = 1/num
End Function
```

(Cálculo do inverso de um número)

Funções – Exemplo

- Elabore uma função que calcule o factorial de um número.

– Fórmula do factorial: $n! = \prod_{k=1}^n k = n \times (n-1) \times \dots \times 1$

- O factorial de um número cresce muito rapidamente, pelo que é necessário usar o tipo de dados não exacto **Double**.

```
Function fact(ByVal n as Integer) as Double
    Dim k as Integer, f as Double
    f = 1
    For k = 2 To n
        f = f * k
    Next
    fact = f                                ` Retorna valor da função
End Function
```

Exemplo de aplicação

- Elabore um programa que calcule combinações.


- Resolução:

– Fórmula: $C_p^n = \frac{n!}{(n-p)!p!}$

Em que **n** é a população e **p** a amostra.

- Vamos utilizar a função desenvolvida atrás para cálculo do factorial.
- Um exemplo de aplicação seria o cálculo das combinações possíveis do Totoloto → C(49, 6).

Resolução I



`txtN` - *TextBox* para introdução da população.

`txtP` - *TextBox* para introdução da amostra.

`txtCom` - *TextBox* para apresentação do resultado.

`cmdcalc` - *CommandButton*
para executar o cálculo

`Option Explicit`

```
Private Sub cmdSair_Click()  
    End  
End Sub
```

↓ (continua)

Resolução II

```
Private Sub cmdCalc_Click()                                ↓ (continuação)  
    Dim n As Integer, p As Integer  
    n = Abs(Val(txtN.Text))  
    p = Abs(Val(txtP.Text))  
    If n > 0 And p > 0 And n >= p Then  
        txtComb.Text = fact(n) / fact(n - p) / fact(p)  
    Else  
        txtComb.Text = ""  
        MsgBox "Erro: dados inválidos!", vbOKOnly  
    End If  
End Sub                                                    A função é chamada 3 vezes.
```



```
Function fact(n As Integer) As Double  
    Dim x As Integer, f As Double  
    f = 1  
    For x = 2 To n  
        f = f * x  
    Next  
    fact = f  
End Function
```

← `fact = f` `fact` contém o resultado da função.

Subrotinas - Sintaxe

• Sintaxe


- Geral

```
Sub <nome>([<argumento1>, ...])  
  <bloco de instruções>  
End Sub
```

- Exemplo

```
Sub Troca(ByRef a As Long, ByRef b As Long)  
  Dim aux As Long  
  aux = a  
  a = b  
  b = aux  
End Sub  
(troca do valor de duas variáveis do tipo Long)
```

Exemplo




```
Option Explicit  
Private Sub cmdOrd_Click()  
  Dim a As Long, b As Long  
  a = Val(txtA.Text)  
  b = Val(txtB.Text)  
  If a > b Then  
    Troca a, b  
  End If  
  txtA.Text = a  
  txtB.Text = b  
End Sub
```

```
Private Sub cmdSair_Click()  
  End  
End Sub  
  
Sub Troca(ByRef x As Long, ByRef y As Long)  
  Dim aux As Long  
  aux = x  
  x = y  
  y = aux  
End Sub
```

Chamada da subrotina. Note-se a ausência de parêntesis.

Exemplo de aplicação

Elabore um programa que permita calcular algumas estatísticas sobre as notas de uma turma: melhor nota e respectivo aluno, pior nota e respectivo aluno e a médias das notas. Pretende-se ainda listar todas as notas lidas. As notas são entre 0 e 20 e o número dos alunos entre 1000 e 2000, inclusive.



Resolução I

```
Option Explicit
Private Sub cmdSair_Click()
    End
End Sub

Private Sub cmdLer_Click()
    Dim na As Integer, med As Single, max As Integer, min As Integer
    Dim melhor As Long, pior As Long

    na = Val(txtNA.Text)
    If na < 1 Then
        MsgBox "Nº de alunos inválido!"
        Exit Sub
    End If

    Estatisticas na, med, max, melhor, min, pior

    txtPA.Text = pior
    txtMA.Text = melhor
    txtMed.Text = med
    txtMin.Text = min
    txtMax.Text = max
End Sub
```

Chamada da subrotina. A subrotina interpreta os argumentos pela ordem em que são inseridos.

Resolução II

```
Sub Estatisticas(ByVal n As Integer, ByRef med As Single, _
    ByRef mNota As Integer, ByRef mAluno As Long, ByRef pNota As Integer, _
    ByRef paluno As Long)
    Dim x As Integer, soma As Integer, nota As Integer, aluno As Long

    pNota = 21      ' valores iniciais
    mNota = -1
    For x = 1 To n
        aluno = LerAluno(x, 1000, 2000)
        nota = LerNota(x, 0, 20)
        soma = soma + nota
        lstNotas.AddItem aluno & " - " & Format(nota, "@@")
        If nota > mNota Then
            mNota = nota
            mAluno = aluno
        End If
        If nota < pNota Then
            pNota = nota
            paluno = aluno
        End If
    Next
    med = soma / n
End Sub
```

Resolução III

```
Function LerNota(ByVal n As Integer, ByVal li As Integer, _
    ByVal ls As Integer) As Single
    Dim nota As Single

    nota = Val(InputBox("Insira a nota n° " & n))
    Do While nota < li Or nota > ls
        nota = Val(InputBox("Valor errado! Insira novamente a nota n° " _
            & n))
    Loop
    LerNota = nota
End Function

Function LerAluno(ByVal n As Integer, ByVal li As Long, _
    ByVal ls As Long) As Long
    Dim aluno As Long

    aluno = Val(InputBox("Insira o número do aluno n° " & n))
    Do While aluno < li Or aluno > ls
        aluno = Val(InputBox("Valor errado! Insira novamente o aluno n° " _
            & n))
    Loop
    LerAluno = aluno
End Function
```

Seleccção da rotina adequada

- Pode ser usada uma função quando:
 - A rotina retorna um só valor e pretende-se utilizar a rotina em expressões.
 - Exemplo: `x = fact(a)/b`
- Nos restantes casos utilizam-se subrotinas.